


Cylon Attack



**Some kind of SPACE INVADERS was always on my mind to try in 1K hires.
Easier said than done. Finally I turned the display and made it a horizontal invader.
Only 1 shot per line is possible, making it tough enough to play.**

```
; Battlestar Galactica :Cylon attack  
; Game 74 in 1K hires for the ZX81.
```

```
? * TORNADO *
```

```
lines      EQU 14  
nrcylon    EQU 16
```

```
size       EQU 25  
exit       EQU #4011
```

```
field      EQU nextlin
```

```
ORG #4009          ;#4009  
DUMP 49161
```

```
; program starts here, both BASIC and machinecode  
basic      EX  AF,AF'          ; delay intrupt,opcode no bit6  
           LD  H,B            ; preset for 48K bug to #40  
           JR  init0          ; continue where room  
  
           DEFB 236,212,28     ; The BASIC  
           DEFB 126           ; fully placed over sysvar  
           DEFB 143,0,18      ; start BASIC=#4009 also MC
```

```
           DEFW last          ; needed by loading  
chadd      DEFW last-1  
xptr       DEFW 0  
stkbot     DEFW last  
stkend     DEFW last  
berg       DEFB 0  
mem        DEFW 0  
           DEFB 128  
  
           DEFB 0,0,0
```

```

; all above reusable AFTER loading

lastk      DEFB 255,255,255      ; used by ZX81
margin     DEFB 55                ; used by ZX81
nxtlin     DEFW basic            ; reusable after load

init0      DEFB 0,0,0,0          ; IX set AFTER lowres screen

          LD    E,L              ; DE now #xx.L

taddr      DEFW 0                ; used by ZX81 on LOAD only
          ; unharmed code

          LD    B,4              ; copy >1K code

frames     DEFB #16+1            ; LD D,n , after LOAD -1
          DEFB #C0                ; highbyte must have bit 7 set
coords     LDIR                  ; DE now #C0.L = H1 + #8000
          ; fix 48K bug before display
prcc       JP    init            ; continue to mainprog

cdfldflag  DEFB 64                ; used by ZX81

; Place ANY code to fill up to #4040
low        LD    A,0
          LD    (BC),A
          DEC   E
          JP    Z,bloop          ; test end of 8 lines reached

cloop      LD    (low+1),A
          LD    A,(DE)
          LD    A,(DE)
          NOP
          LD    A,(DE)            ; get show/noshow fire
          LD    (BC),A            ; set fire signal
          DEC   DE
          LD    A,(DE)            ; get characterline
          LD    R,A                ; set lowbyte of hires
          JP    (IX)              ; show line through stack

; some lowres, HR must start AFTER #403F, but before #4070
hr         LD    HL,lowres+#8000  ; the lowres display
          LD    BC,#251            ; minimum needed #11
          LD    A,#1E
          LD    I,A
          LD    A,#FB
          CALL  #2B5              ; show lowres screen

; the hr part
          LD    B,7
hr0        DJNZ  hr0              ; sync hires with lowres
          LD    (exit+1),SP
          LD    SP,scrstack
          LD    HL,low            ; back from display
          LD    A,H
          LD    IX,lbuf-size-1+#8000
          LD    I,A                ; set highbyte display

; 207 tstates delay cheaper than line
; get fireposistion somehow

bloop      LD    DE,size+1        ; get fireposition
          ADD   IX,DE

```

```

        INC    DE
        LD     A,(HL)
        LD     B,4
linedelay LD     DE,14+16384
        DJNZ   linedelay

        DEFB   #DD
        LD     A,L
        INC    A
        JR     Z,exit           ; test end of screen
        POP    BC
        LD     A,(BC)
        CP     #E9             ; is fire on end of line
        JP     NZ,notbc
notbc    DEFB   1               ; if so, make BC in ROM
        INC    E               ; use BC from stack
        DEC    DE             ; 10 tstates harmless sync code
        JR     cloop

delay    LD     A,252
        LD     HL,frames       ; standard delay routine
        ADD    A,(HL)
wfr      CP     (HL)
        JR     NZ,wfr
        RET

11       DEFB   224
        DEFB   120,120,120,120,120,120
        DEFB   120,120,120,120,120,120
        DEFB   120,120,120,120,120,120
        DEFB   120,120,120,120,120,120

12       DEFB   144
        DEFB   12,12,12,12,12,12
        DEFB   12,12,12,12,12,12
        DEFB   12,12,12,12,12,12
        DEFB   12,12,12,12,12,12

13       DEFB   255
        DEFB   126,126,126,126,126,126
        DEFB   126,126,126,126,126,126
        DEFB   126,126,126,126,126,126
        DEFB   126,126,126,126,126,126

eog      LD     HL,score-1      ; your score
        LD     DE,hiscore-1    ; hiscore
        LD     BC,5            ; size 4
fihi     INC    HL
        INC    DE
        DEC    C               ; when C=0
        LD     A,(DE)          ; (de) = #76
        CP     (HL)            ; (hl) = 0
        JR     Z,fihi          ; so not same
        CALL   C,#19F9         ; and no hiscore

start    LD     A,(lastk)       ; game over, wait for
        SUB    %10111111       ; newline
        JR     NZ,start

clsco    LD     HL,score
        LD     (HL),28         ; reset old score
        INC    HL
        CP     (HL)
        JR     NZ,clsco

```

```

        LD    A,32
        LD    (lives),A

dead    LD    HL,lives
        DEC   (HL)
        LD    A,(HL)
        CP    28
        JR    Z,eog

cls     LD    HL,lbuf-1          ; end of 1K RAM end of lbufs
        LD    SP,HL
        LD    DE,scrstack
        LD    B,8                ; 8 cylons on a row

clbuf   LD    A,(HL)
        CP    #E9
        JR    NZ,cllbuf
        LD    A,L
        LD    (DE),A
        INC   DE
        INC   DE
        DEC   B
        LD    C,258-size        ; 2! rows of cylons
        INC   HL

cllbuf  LD    (HL),#40           ; erase visible character
        LD    A,H               ; get value > 63
        CP    B
        JR    C,cldone          ; test all cylons on row done
        CP    C
        JR    C,cldone          ; test rows with cylons done
        LD    (HL),C            ; show cylon

cldone  INC    C
        INC   HL                ; go to next field
        LD    A,H
        SUB   #44
        JR    NZ,clbuf          ; test end of screen reached
        LD    A,lines
        LD    (cylonline+1),A
        LD    A,size-2
        LD    (cylpos+1),A
        LD    A,nrcylon
        LD    (lastk-1),A

        LD    A,200
        CALL  delay+2            ; delay before new screen starts

player  LD    B,5                ; position of player

playloop LD    HL,scrstack
        LD    C,14

fireup  LD    E,(HL)
        INC   HL
        LD    D,(HL)
        LD    A,(DE)
        ADD   A,A
        JR    C,skipfire
        ADD   A,A
        JR    NC,addp
        INC   DE
; test visible, then hit
        LD    A,(DE)
        CP    #40
        JR    Z,nohit
        JR    NC,nohit

```

```

addp      LD    A,#40
          LD    (DE),A
; score here
          PUSH HL
          LD    HL,score+4
          DEFB #3A
ten        LD    (HL),28
          DEC   HL
          INC   (HL)
          LD    A,(HL)
          CP    38
          JR    Z,ten
          LD    HL,lastk-1
          DEC   (HL)
          POP   HL
          JR    Z,cls

findend    INC   DE
          LD    A,(DE)
          ADD   A,A
          JR    NC,findend

nohit      LD    (HL),D
          DEC   HL
          LD    (HL),E
          INC   HL
skipfire   INC   HL
          DEC   C
          JR    NZ,fireup

          CALL  field

          LD    (HL),C          ; show player
          PUSH HL
          PUSH BC

cylonline  LD    B,14
cylpos     LD    C,0
          CALL  field
cont       LD    A,#40
          CP    (HL)
          INC   HL
          JR    NZ,leftstep
          CP    (HL)
          JR    NZ,leftstep

; up/down move
clp        PUSH BC
          CALL  field

          LD    A,B
          ADD   A,250
          LD    DE,size+1
          JR    NC,blp
          LD    DE,65535-size

blp        PUSH HL
          ADD   HL,DE
          LD    A,(HL)
          LD    (HL),#40
          POP   BC
          LD    (BC),A

cntlp      LD    A,lines-1

```

```

DEC    A
JR     NZ,wcnt
LD     A,lines-1
wcnt   LD     (cntlp+1),A
JR     NZ,blp

POP    BC
LD     A,(cylpos+1)
CP     C
INC    BC

JR     Z,clp
JR     show

scrstack DEFW lbuf-1
        DEFW 1*size+lbuf
        DEFW 2*size+lbuf+1
        DEFW 3*size+lbuf+2
        DEFW 4*size+lbuf+3
        DEFW 5*size+lbuf+4
        DEFW 6*size+lbuf+5
        DEFW 7*size+lbuf+6
        DEFW 8*size+lbuf+7
        DEFW 9*size+lbuf+8
        DEFW 10*size+lbuf+9
        DEFW 11*size+lbuf+10
        DEFW 12*size+lbuf+11
        DEFW 13*size+lbuf+12
        DEFW 14*size+lbuf+13

leftstep LD     DE,lbuf+1
        LD     A,15
        LD     HL,cylonline+1
        SUB    (HL)
        LD     (HL),A
        LD     HL,cylpos+1
        DEC    (HL)

all      LD     H,D
        LD     L,E
        LD     BC,size-2
        INC    HL
        LD     A,C
        CP     (HL)
        LDIR
        LD     A,#40
        LD     (DE),A
        JP     NC,dead
        INC    DE
        INC    DE
        INC    E
        JR     NZ,all

show     CALL    delay                ; show screen

        LD     BC,(lastk)
        LD     A,C
        INC    A
        CALL    NZ,#7BD
        POP    BC

        POP    HL
        LD     (HL),#40                ; erase player

```

```

CP      25                                ; "p"
JR      NZ,updown
CALL    field
INC      HL                                ; HL now start of fire
LD      DE,scrstack-2
LD      A,B
ADD      A,A
ADD      A,E
LD      E,A
EX      DE,HL
PUSH     DE
LD      E,(HL)
INC      HL
LD      D,(HL)
LD      A,(DE)
POP      DE
CP      #E9
JR      NZ,nofire
LD      (HL),D
DEC      HL
LD      (HL),E
nofire   LD      A,H
updown   LD      E,B
CP      10                                ; "q"
JR      NZ,key2
DEC      B
key2     CP      5                          ; "a"
JR      NZ,testb
INC      B
testb    LD      A,B
DEC      A
CP      14
JR      C,okmove
LD      B,E
okmove   JP      playloop

x        EQU     101
lowres   DEFB    118
score    DEFB    28,28,28,28,0
          DEFB    "C"+x,"Y"+x,"L"+x,"O"+x,"N"+x,128
          DEFB    "A"+x,"T"+x,"T"+x,"A"+x,"C"+x,"K"+x,0
lives    DEFB    28,0
hiscore  DEFB    28,28,35,32
          DEFB    118

lbsize   EQU     lines*size+lines
mark     EQU     #43FF-lbsize

space    EQU     mark-$$-1
          DEFS    space

lbuff    JP      (HL)
          DEFW    0,0,0,0,0,0,0,0,0,0,0,0,0
          DEFB    0
          JP      (HL)

; during game this is screen. After loading this
; part of text and wait for newline is executed
init     LD      SP,lbuff-1                ; move SP
          LD      HL,scr2
          LD      (basic+3),HL

          LD      HL,cent

```

```

LD      C,1

w2      SET  7,(HL)
w21     LD   A,L
        ADD  A,C
        LD   L,A
        BIT  7,(HL)
        JR   NZ,clbit
        XOR  A
        SUB  C
        LD   C,A
        JR   w21
clbit   RES  7,(HL)
        EX   DE,HL
        CALL delay
        EX   DE,HL

        LD   A,(lastk)
        SUB  %10111111
        JR   NZ,w2
        EX   AF,AF'           ; delay intrupt for copy time

        LD   IX,hr

        LD   HL,detab          ; table for display lines
        LD   DE,#4001          ; set over sysvar
        LD   C,33              ; exitroutine HR too on sysvar
        LDIR

        LD   HL,fieldc         ; simple RND-routine
        LD   DE,nxtlin         ; also set over sysvar to
        LD   C,18              ; save bytes in codeable
        LDIR                   ; memory

        LD   HL,lbuf           ; first line of display
        LD   DE,init           ; second line of display
        LD   BC,#4400-init     ; fill rest of memory
        SCF
        JP   start-3           ; with LDIR from hiscore

fieldc  PUSH  BC
        LD   A,E
        LD   HL,lbuf-size-1
        LD   DE,size+1
lb1     ADD  HL,DE
        DEC  B
        DEFB #DA               ; pos 11 JP C
        DEFW 65535
        JR   NZ,lb1
        ADD  HL,BC
        POP  BC
        RET

detab   DEFB 11*256/256-2,64
        DEFB 12*256/256-2,64
        DEFB 13*256/256-2,64
        DEFB 13*256/256-2,0
        DEFB 13*256/256-2,64
        DEFB 12*256/256-2,64
        DEFB 11*256/256-2,64
        DEFW 0

; fixed end of HR-routine

```



```

exitc    LD    SP,0
         CALL #292                ; back from intrupt
         CALL #220
         LD    IX,hr
         JP    #2A4

n        EQU   27

scr2     DEFB  118

         DEFB  "C"+x,"Y"+x,"L"+x,"O"+x,"N"+x,128
         DEFB  "A"+x,"T"+x,"T"+x,"A"+x,"C"+x,"K"+x,118,118

         DEFB  "B"+x,"A"+x,"L"+x,"T"+x,"A"+x,"R"+x,0
         DEFB  "C"-n,"E"-n,"N"-n,"T"-n,"U"-n,"R"-n

         DEFB  "I"-n,"O"-n,"N"-n,26,118

         DEFB  "A"-n,"T"-n,"T"-n,"A"-n,"C"-n,"K"-n,0
         DEFB  "T"-n,"H"-n,"E"-n,0,"G"-n,"A"-n,"L"-n
         DEFB  "A"-n,"C"-n,"T"-n,"I"-n,"C"-n,"A"-n,0

         DEFB  "W"-n,"I"-n,"T"-n,"H"-n,118
         DEFB  29,34,0,"R"-n,"A"-n,"I"-n,"D"-n,"E"-n
         DEFB  "R"-n,"S"-n,0,"I"-n,"N"-n,0,"W"-n,"A"-n
         DEFB  "V"-n,"E"-n,"S"-n,118

         DEFB  "A"-n,"N"-n,"D"-n,0,"Y"-n,"O"-n,"U"-n,0
         DEFB  "W"-n,"I"-n,"L"-n,"L"-n,0
         DEFB  "S"-n,"U"-n,"C"-n,"C"-n,"E"-n,"E"-n,"D"-n,118
         DEFB  118

cent     DEFB  "C"+x,"E"+x,"N"+x,"T"+x,"U"+x,"R"+x
         DEFB  "I"+x,"O"+x,"N"+x,0

         DEFB  "B"-n,"Y"-n,0,"Y"-n,"O"-n,"U"-n,"R"-n,0
         DEFB  "C"-n,"O"-n,"M"-n,"M"-n,"A"-n,"N"-n
         DEFB  "D"-n,118

         JP    (HL)

vars     DEFB  128
last     EQU   $

```